Karolína Kundrátová

NURBS REPREZENTACE KŘIVEK V MAPLE

Abstrakt

Parametrizace křivek jako NURBS (tj. neuniformní racionální B-spliny) patří k moderním postupům geometrického modelování. V příspěvku je uveden obecný výpočet takovéto reprezentace křivek pomocí matematického programu Maple.

Klíčová slova

NURBS křivka, Maple.

1 Úvod

Jedním ze základních úkolů, které jsou řešeny v geometrickém modelování, je otázka, jak množinou bodů v rovině či prostoru proložit křivku vhodného tvaru. Víme, že takové křivky můžeme rozdělit do dvou skupin: na ty, které procházejí danými body (interpolační křivky), a ty, které sledují tvar lomené čáry spojující řídicí body, tzv. řídicí polygon (aproximační křivky). Existuje mnoho metod, jak interpolaci či aproximaci provést. Článek se zabývá aproximací pomocí NURBS.

NURBS je velmi používaným termínem moderní počítačové geometrie. Uveďme ve stručnosti, co zkratka znamená. NURBS jsou neuniformní racionální B-spliny. B-spliny jsou segmentované křivky složené z oblouků známých Bézierových křivek a splňující podmínky C^2 spojitosti. Racionální B-spliny se od polynomiálních křivek liší v tom, že každému řídicímu bodu je přiřazen parametr vypovídající o jeho vlivu na tvar křivky. Parametru říkáme váha a nabývá nezáporných reálných hodnot. Čím větší je váha jednoho bodu oproti váhám ostatních bodů, tím větší vliv má bod na tvar křivky (tím více je křivka „přitahována“ k tomuto bodu). Racionální B-spliny jsou parametrizovány pomocí racionálních funkcí. O neuniformních racionálních B-splinech hovoříme, když vzdálenosti mezi body, ve kterých dojde k napojení segmentů (těmto bodům říkáme uzly; posloupnosti, kterou tvoří, pak uzlový vektor), nejsou stejné. Vzdáleností uzlů zde myslíme rozdíl hodnot parametru u NURBS křivky $C(u)$ v uzlech. Když jsou vzdálenosti uzlů stejné, jedná se o uniformní racionální B-spliny.

V článku je prezentován výpočet NURBS křivky podle její definice pomocí matematického programu Maple. Tento výpočet byl sestaven pro potřeby dalšího zkoumání NURBS křivek a byl využit při výuce na FSI ČVUT v Praze v rámci volitelného semináře Geometrie pro CAD (viz `http://marian.fsik.cvut.cz/~linkeova` pod odkazem `Geometrie pro CAD`.

2 NURBS křivky

V dalším textu se budu odvolávat na definici uzlů a uzlového vektoru (1), definici B-spline bázových funkcí (2) a NURBS křivky (3), které jsou zařazeny v části 2 článku [1]. Zájemcům o problematiku doporučuji knihu [2] nebo didakticky vhodně zpracované internetové přednášky [3]. Přechodem mezi NURBS reprezentací a tradičními parametrizacemi křivek (Fergusonova kubika, Bézierova křivka, Coonsova kubika) se zabývá článek [1].

2.1 Výpočet NURBS křivek pomocí Maple

K tomu, abychom mohli zkoumat vlastnosti NURBS křivek a s křivkami dále pracovat, jsme sestavili výpočet v matematickém programu Maple. Použili jsme verzi Maple 8.00, viz [4]. Studenti výše zmíněného semináře měli za úkol ručně spočítat NURBS křivky druhého a třetího stupně určené čtyřmi řídicími body. Nejtěžší pro ně bylo vyrovnat se s tím, že bázové funkce i samotné křivky jsou jinak definované na různých intervalech uzlového vektoru. V Maplu si s tímto faktem poradíme použitím příkazu `piecewise` (viz níže).

Následující výpočet je sestaven pro rovinnou NURBS křivku C stupně p určenou $n+1$ řídicími body $P[i]$, kde $i = 0, \ldots, n$. Každému bodu $P[i]$ je přirazena váha $w[i]$. Uzlový vektor je označen U a ve výpočtu je uniformní.

Parametry, které můžeme měnit, jsou: počet řídicích bodů $n+1$, poloha a váhy řídicích bodů $P[i]$, stupeň křivky p a uzlový vektor U. Vliv změny těchto tvarovacích parametrů na tvar křivky si ukážeme níže v části 2.2. Pro výpočet prostorové NURBS křivky můžeme použít stejný program, provedeme-li drobnou úpravu, a sice přidáme-li třetí souřadnice řídicích bodů a přeindexujeme pole s váhami bodů.

Index koncového řídicího bodu je označen n; počet řídicích bodů je *n+1.*

```
>  n:=4:
```

Řídicí body křivky jsou $P[i];\ i=\{0,\ 1,\ \ldots,\ n\}$, přičemž první dvě složky vektoru jsou souřadnice (pracujeme v rovině), třetí složka je váha.

```
>  P[0]:=([0,0,1]):
>  P[1]:=([1,3,1]): P[2]:=([3,1,12]): P[3]:=([4,5,1]):
>  P[4]:=([6,-1,1]):
```

Váhy bodů přepíšeme do pole *w,* kde *w[i]* je váha bodu *P[i].*

```
>  for i from 0 to n do
>     w[i]:=P[i][3]
>  od:
```

p je stupeň křivky.

```
>  p:=2:
```

nops_U je počet složek uzlového vektoru.

```
>  nops_U:=n+p+2:
```

Výpočet uzlů tak, aby uzlový vektor byl uniformní – stejně dlouhé intervaly – a křivka interpolovala krajní řídicí body – z teorie NURBSů víme, že uzlový vektor musí mít prvních *p+1* složek rovných 0 a posledních *p+1* složek rovných 1.

```
>  for i from 1 to p+1 do
>     U[i]:=0
>  od:
>  for i from p+2 to n+1 do
>     U[i]:=U[i-1]+1/(n-p+1)
>  od:
>  for i from n+2 to nops_U do
>     U[i]:=1
>  od:
```

Přeindexování uzlů tak, aby první uzel měl index *0* - nové pole uzlů nazvané *uu.*

```
>  for i from 1 to nops_U do    uu[i-1]:=U[i]
>  od:
```

Nulté bázové funkce *N[i,0]* dle definice 2; je třeba definovat segmentovaně (jiný předpis pro různé intervaly).

```
>  for i from 0 to nops_U-2 do
>     if (uu[i+1]-uu[i]<>0) then
>        if i<>n
>           then N[i,0]:=simplify(piecewise(u>=uu[i]
>                 and u<uu[i+1],1,0)):
>           else N[i,0]:=simplify(piecewise(u>=uu[i] and
>                 u<=uu[i+1],1,0)):
>        fi:
>        else N[i,0]:=0
>     fi
>  od:
```

Bázové funkce *N[i,s]* stupně prvního až p-tého dle definice 2.

```
>   for s from 1 to p do
>     for i from 0 to nops_U-2-p do
>       if (uu[i+s]-uu[i]=0)
>         then a[i,s]:=0
>         else a[i,s]:=(u-uu[i])/(uu[i+s]-uu[i])*N[i,s-1]
>       fi;
>       if (uu[i+s+1]-uu[i+1]=0)
>         then b[i,s]:=0
>         else b[i,s]:=(uu[i+s+1]-u)/(uu[i+s+1]-uu[i+1])
>                 *N[i+1,s-1]
>       fi;
>       N[i,s]:=simplify(a[i,s]+b[i,s])
>     od
>   od:
```

Racionální bázové funkce *R[i,p]* stupně p dle definice 3.

```
>   for i from 0 to n do
>     R[i,p]:=simplify(N[i,p]*w[i]/(sum('N[j,p]*w[j]',
>            'j'=0..n)))
>   od:
```

NURBS křivka C dle definice 3.

```
>   for k from 1 to 2 do
>     C[k]:=simplify(sum('R[i,p]*P[i][k]','i'=0..n))
>   od;
```

$$C_1 = \begin{cases} \mathit{undefined} & u < 0 \\ \dfrac{3\,u\,(4+99\,u)}{99\,u^2+2} & u < \dfrac{1}{3} \\ \dfrac{603\,u^2-612\,u+100}{198\,u^2-198\,u+31} & u < \dfrac{2}{3} \\ \dfrac{12\,(27\,u^2-52\,u+26)}{99\,u^2-198\,u+101} & u \le 1 \\ \mathit{undefined} & 1 < u \end{cases} \quad C_2 = \begin{cases} \mathit{undefined} & u < 0 \\ \dfrac{9\,u\,(4+3\,u)}{99\,u^2+2} & u < \dfrac{1}{3} \\ \dfrac{144\,u^2-150\,u+19}{198\,u^2-198\,u+31} & u < \dfrac{2}{3} \\ -\dfrac{45\,u^2-18\,u-25}{99\,u^2-198\,u+101} & u \le 1 \\ \mathit{undefined} & 1 < u \end{cases}$$

Zde je třeba upřesnit, že program Maple nabízí práci s knihovnou `CurveFitting`, která obsahuje příkazy pro výpočet křivek určených množinou řídicích bodů. Samotné NURBS křivky pomocí příkazů knihovny počítat nemůžeme, ale např. výpočet, který jsme právě komentovali, ovšem pro stejné váhy všech řídicích bodů (tj. výpočet B-spline křivky), můžeme provést pomocí příkazu `BSplineCurve`, jehož skladbu ukazují následující řádky (parametry byly zvoleny stejně jako ve výpočtu uvedeném výše):

```
>   with(CurveFitting):
>   simplify(BSplineCurve([0,1,3,4,6],[0,3,1,5,-1],
>   u,order=3,knots=[0,0,0,1/3,2/3,1,1,1]));
```

Pro potřeby výuky a hlubší pochopení celého problému je ale vlastní výpočet, ve kterém je názorně užitá definice NURBS křivek, vhodnější.

2.2 Příklady NURBS křivek

K tomu, abychom mohli vhodně demonstrovat vliv změny volitelných parametrů (váha bodu, stupeň křivky, uzlový vektor) na tvar NURBS křivky, jsme zvolili křivku určenou 16 řídicími body. Z důvodu nedostatku místa nebudeme uvádět matematické reprezentace jednotlivých křivek, ale uvedeme jejich grafy. Hodnoty parametrů jednotlivých křivek jsou upřesněny pod každým obrázkem.



Obr. 1: Křivky 12. stupně. Odleva: $w_i = 1$, uniformní U. $w_4 = 20$, ostatní $w_i = 1$, uniformní U. $w_i = 1$, neuniformní $U = \{\underbrace{0, \ldots, 0}_{13\times}, \frac{1}{5}, \frac{2}{3}, \frac{7}{8}, \underbrace{1, \ldots, 1}_{13\times}\}$



Obr. 2: Křivky s $w_i = 1$ a uniformním U. Změna stupně p (odleva): $p = 15,\ p = 5,\ p = 2$. Čím nižší stupeň křivky, tím věrněji křivka sleduje řídicí polygon.

Poděkování

Článek vznikl za podpory projektu CTU 0513112: NURBS reprezentace křivek a ploch v MAPLE.

Literatura

[1] I. Linkeová: *Speciální případy NURBS reprezentace,* Sborník konference CGC 2005, Janov, 2005.

[2] L. Piegl, W. Tiller: *The NURBS Book,* Springer, Londýn, 1995.

[3] Ch.-K. Shene: *www.cs.mtu.edu/˜shene/COURSES/cs3621/NOTES*

[4] *Maple 8.00 – help programu Maple.*